/*
 *  Licensed to the Apache Software Foundation (ASF) under one
 *  or more contributor license agreements.  See the NOTICE file
 *  distributed with this work for additional information
 *  regarding copyright ownership.  The ASF licenses this file
 *  to you under the Apache License, Version 2.0 (the
 *  "License"); you may not use this file except in compliance
 *  with the License.  You may obtain a copy of the License at
 *
 *    https://www.apache.org/licenses/LICENSE-2.0
 *
 *  Unless required by applicable law or agreed to in writing,
 *  software distributed under the License is distributed on an
 *  "AS IS" BASIS, WITHOUT WARRANTIES OR CONDITIONS OF ANY
 *  KIND, either express or implied.  See the License for the
 *  specific language governing permissions and limitations
 *  under the License.
 */

import java.time.Instant
import java.time.ZoneOffset
import java.time.format.DateTimeFormatter

plugins {
    id 'java-library'
    id 'org.apache.grails.gradle.grails-plugin'
}

version =  projectVersion
group = 'org.apache.grails'

def buildInstant = java.util.Optional.ofNullable(System.getenv('SOURCE_DATE_EPOCH'))
        .filter(s -> !s.isEmpty())
        .map(Long::parseLong)
        .map(Instant::ofEpochSecond)
        .orElseGet(Instant::now) as Instant
ext {
    formattedBuildDate = DateTimeFormatter.ISO_INSTANT.format(buildInstant)
    buildDate = buildInstant.atZone(ZoneOffset.UTC) // for reproducible builds
    isCiBuild = System.getenv().containsKey('CI')
}

repositories {
    mavenCentral()
    maven { url = 'https://repo.grails.org/grails/restricted' }
    maven {
        url = 'https://repository.apache.org/content/groups/snapshots'
        content {
            includeVersionByRegex('org[.]apache[.]grails.*', '.*', '.*SNAPSHOT')
        }
    }
    maven {
        url = 'https://repository.apache.org/content/groups/staging'
        content {
            includeGroupByRegex('org[.]apache[.]grails.*')
        }
    }
}

grails {
    springDependencyManagement = false
}

dependencies {

    implementation platform("org.apache.grails:grails-bom:$grailsVersion")

    api "org.quartz-scheduler:quartz", {
        // api: AdaptableJobFactory, InterruptableJob, JobExecutionContext, JobExecutionException,
        //      UnableToInterruptJobException, TriggerFiredBundle,
        // runtime: @DisallowConcurrentExecution(runtime), @PersistJobDataAfterExecution(runtime)
    }

    compileOnly 'org.apache.grails:grails-dependencies-starter-web'

    testImplementation 'org.apache.grails:grails-dependencies-test'
}

apply {
    from layout.projectDirectory.file('gradle/docs-config.gradle')
    from layout.projectDirectory.file('gradle/java-config.gradle')
    from layout.projectDirectory.file('gradle/publish-config.gradle')
    from layout.projectDirectory.file('gradle/test-config.gradle')
    from layout.projectDirectory.file('gradle/rat-root-config.gradle')
    from layout.projectDirectory.file('gradle/reproducible-config.gradle')
}